# Packager: Justin "juster" Davis <jrcd83@gmail.com>
# $Id: PKGBUILD 180359 2013-03-20 16:53:13Z eric $

pkgname=perl-www-robotrules
pkgver=6.02
pkgrel=1
pkgdesc="database of robots.txt-derived permissions"
arch=(any)
license=(PerlArtistic GPL)
options=(!emptydirs)
depends=('perl>=5.8.1' 'perl-uri>=1.10')
conflicts=('perl-libwww<6')
url=https://metacpan.org/release/WWW-RobotRules
source=("http://search.cpan.org/CPAN/authors/id/G/GA/GAAS/WWW-RobotRules-${pkgver}.tar.gz")
md5sums=(b7186e8b8b3701e70c22abf430742403)
sha512sums=(781a2fc90e4efed901a91ecf4f2aaac2684409c6a75a8434ff140654838bb05d11f8fa760642e20eee96450d3ced6815a4dab11a4245bc8120c69ca68ab49e44)
_distdir="${srcdir}/WWW-RobotRules-${pkgver}"

build()
(
    export PERL_MM_USE_DEFAULT=1 PERL5LIB=""                 \
      PERL_AUTOINSTALL=--skipdeps                            \
      PERL_MM_OPT="INSTALLDIRS=vendor"     \
      PERL_MB_OPT="--installdirs vendor" \
      MODULEBUILDRC=/dev/null

    cd "$_distdir"
    /usr/bin/perl Makefile.PL
    make
)

check()
(
    export PERL_MM_USE_DEFAULT=1 PERL5LIB=""
    cd "$_distdir"
    make test
)

package() {
    cd "$_distdir"
    make DESTDIR="$pkgdir" install
}

# Local Variables:
# mode: shell-script
# sh-basic-offset: 2
# End:
# vim:set ts=2 sw=2 et:
